query: embed.html### A Query to see the % completeness of wikidata translations of Labels, Descriptions and Subclass-labels in the Europeana Migration Vocabulary project
  ## Wikidata homepage: https://www.wikidata.org/wiki/User:Wittylama/Migration_vocabulary
  ## Europeana homepage: https://pro.europeana.eu/page/europeana-migration
  
  #defaultView:Table
  SELECT ?languageLabel (round(100*SUM(?label_exists)/SUM(1)) AS ?Labels_percent_translated) (round(100*SUM(?desc_exists)/SUM(1)) AS ?Descriptions_percent_translated) (round(100*SUM(?classlabel_exists)/SUM(1)) AS ?Class_lables_percent_translated) WHERE {
  values ?item {wd:Q177626 wd:Q131288 wd:Q187668 wd:Q1932461 wd:Q12547146 wd:Q706601 wd:Q473741 wd:Q15320003 wd:Q280958 wd:Q2608153 wd:Q6013750 wd:Q4354288 wd:Q7317170 wd:Q837556 wd:Q379693 wd:Q133004 wd:Q705818 wd:Q181784 wd:Q8463 wd:Q7167 wd:Q11376059 wd:Q59816 wd:Q11409 wd:Q41397 wd:Q230533 wd:Q154278 wd:Q166005 wd:Q8461 wd:Q8458 wd:Q79913 wd:Q1338878 wd:Q192989 wd:Q1368367 wd:Q16513600 wd:Q36192 wd:Q18344718 wd:Q912512 wd:Q8434 wd:Q7181 wd:Q1397926 wd:Q131572 wd:Q2382319 wd:Q283509 wd:Q12765670 wd:Q564811 wd:Q128393 wd:Q133346 wd:Q185598 wd:Q4006 wd:Q170404 wd:Q1351807 wd:Q196756 wd:Q1522225 wd:Q43812 wd:Q2918584 wd:Q61509 wd:Q5843 wd:Q865054 wd:Q200814 wd:Q3172759 wd:Q367293}
  values ?language {wd:Q7918 wd:Q6654 wd:Q9056 wd:Q9035 wd:Q7411 wd:Q9072 wd:Q1860 wd:Q1412 wd:Q150 wd:Q188 wd:Q9129 wd:Q9067 wd:Q9142 wd:Q652 wd:Q9078 wd:Q9083 wd:Q9166 wd:Q809 wd:Q5146 wd:Q7913 wd:Q9027 wd:Q9063 wd:Q9058 wd:Q1321 wd:Q7026 wd:Q25167 wd:Q8785 wd:Q8798 wd:Q25164 wd:Q33947 wd:Q8752 wd:Q143 wd:Q13248 wd:Q12175 wd:Q9296 wd:Q8748 wd:Q9299 wd:Q12107 wd:Q9309}
  ?language wdt:P424 ?languagecode .  # and their Wikimedia language codes
  OPTIONAL {
    ?item rdfs:label ?item_label .         # and match the item label language...
    FILTER(?languagecode=lang(?item_label))    # ...with the language
    BIND(IF(BOUND(?item_label),1,0) AS ?label_exists) # count the number of existing labels
  }
  OPTIONAL {
    ?item schema:description ?item_desc.          # match the item description language...
    FILTER(?languagecode=lang(?item_desc))    # ...with the language
    BIND(IF(BOUND(?item_desc),1,0) AS ?desc_exists) # count the number of existing descriptions
  }
  OPTIONAL {
    ?item wdt:P279 ?subclassof.          #  get the item subclass ...
    ?subclassof rdfs:label ?subclasslabel .# so you can get its subclassLabel
    FILTER(?languagecode=lang(?subclasslabel))    # match the label with the language
    BIND(IF(BOUND(?subclasslabel),1,0) AS ?classlabel_exists) # count the number of existing subclasslabels
  } # be aware that a vocabulary term will be seen as having a 'completed' subclasslabel if there is at least one subclasslabel that is translated if a subclasslabel exists
  SERVICE wikibase:label { bd:serviceParam wikibase:language "en". }  # change "en" to your preferred language
  }
  GROUP BY ?languageLabel
  ORDER BY DESC(?Labels_percent_translated)